chcp 65001 > nul
@echo off
REM 获取当前脚本文件的绝对路径，并从中提取目录部分
set "scriptDir=%~dp0"
set "exeDir=%scriptDir%exe\"
set "lockedDir=%scriptDir%locked\"

REM 检查 exe 目录是否存在
if not exist "%exeDir%" (
    echo exe 目录不存在！
    pause
    exit /b
)

REM 检查 locked 目录是否存在
if not exist "%lockedDir%" (
    echo locked 目录不存在！
    pause
    exit /b
)

REM 查找 exe 目录中的第一个 .exe 文件
for /f "delims=" %%i in ('dir /a-d /b /s "%exeDir%\*.exe" 2^>nul') do (
    set "firstExe=%%i"
    goto :foundExe
)

:foundExe
if not defined firstExe (
    echo exe 目录中没有找到 .exe 文件！
    pause
    exit /b
)

echo 使用 %firstExe% 进行解锁

REM 使用找到的 .exe 文件打开 locked 目录中的所有文件
for /f "delims=" %%j in ('dir /a-d /b /s "%lockedDir%\*.*" 2^>nul') do (
    "%firstExe%" -o %scriptDir%/unlocked "%%j"
    REM echo 已使用 %firstExe% 打开 %%j
)

echo 所有文件已解锁！
pause
exit /b